A
Homogeneous Solution of the Einstein-Maxwell Equations
Synopsis
We exhibit and analyze a homogeneous spacetime whose source is a pure radiation
electromagnetic field [1]. It was previously believed that this spacetime is the sole example of
a homogeneous pure radiation solution of the Einstein equations which admits no
electromagnetic field (see [2] and references therein). Here we correct this error in the
literature by explicitly displaying the electromagnetic source. This result implies that all
homogeneous pure radiation spacetimes satisfy the Einstein-Maxwell equations.
In this worksheet we shall
1. Display the metric and electromagnetic field and verify that the Einstein and Maxwell
equations are satisfied.
2. Exhibit some geometric properties of the solution: the electromagnetic field is null, the
Petrov type is N, and the spacetime is a non-reductive homogeneous space.

1. The Einstein-Maxwell solution
Load in the required packages.
> with(DifferentialGeometry): with(Tensor): with(GroupActions):
with(LieAlgebras): with(Library):
Define a chart and construct the metric. This is (12.36) in "Exact Solutions to Einstein's
Field Equations" by Stephani, et al. We can obtain this metric from the DG library using the
Retrieve command.
> DGsetup([u,v,x,y], M);

frame name: M
> g := Retrieve("Stephani", 1, [12, 36, 1], manifoldname = M,
output = ["Fields"])[1];
g := K2 e2 _r x du du C du dv C dv du C dx dx C dy dy

(1.1)

(1.2)

2

Compute the Einstein tensor. It is of the form G = F vv 5 vv corresponding to an
energy-momentum tensor of pure radiation type in terms of the null vector field vv.
M > G := EinsteinTensor(g);
G := 4 _r 2 e2 _r x D_v D_v

(1.3)

The following 2-form is the electromagnetic field strength. The function f u is arbitrary.

M > F := evalDG(2*exp(_rho*x)*rho*cos(_rho*y + f(u))*du &w dx 2*exp(_rho*x)*rho*sin(_rho*y + f(u))*du &w dy);
F := 2 e_r x r cos _r y C f u du Y dx K 2 e_r x r sin _r y C f u du Y dy

(1.4)

We verify the Einstein equations are satisfied by computing the electromagnetic energymomentum tensor,
M > T := EnergyMomentumTensor("Electromagnetic", g, F);
2
T := 4 e2 _r x r D_v D_v

(1.5)

and then verifying that it equals the Einstein tensor:
M > evalDG(G - T);

2

K4 e2 _r x r K _r 2 D_v D_v

(1.6)

Next we verify that the 2-form F satisfies the source-free Maxwell equations in the
spacetime defined by g.
M > MatterFieldEquations("Electromagnetic", g, F);
0 D_u, 0 du Y dv Y dx

(1.7)

2. Properties of the solution
Here we exhibit some geometrical properties of this electrovac spacetime.

The electromagnetic field is null.
We begin by establishing the electromagnetic field is null: its two scalar invariants vanish.
The first invariant is the complete contraction of F with itself.
M > TensorInnerProduct(g, F, F);
0

(2.1)

The second invariant is the Hodge dual of F and its complete contraction with F.
M > Fdual := HodgeStar(g, F, detmetric = -1);
Fdual := 2 e_r x r sin _r y C f u du Y dx C 2 e_r x r cos _r y C f u
M > TensorInnerProduct(g, F, Fdual);
0

du Y dy

(2.2)
(2.3)

The Petrov type is N.
The Petrov type is easily computed.
M > PetrovType(g);

"N"

(2.4)

There are 2 covariantly constant vector fields.
The space of covariantly constant vector fields is two-dimensional.
M > CovariantlyConstantTensors(g, [D_u, D_v, D_x, D_y]);
D_y, D_v

(2.5)

The isometry group is 5 dimensional and the electromagnetic field is noninheriting.
We find a basis for the Killing vectors, which generate a five dimensional group of
isometries.
M > KV := KillingVectors(g);
KV := Ky D_v C u D_y, D_y, K_r u D_u C _r v D_v C D_x, D_u, D_v

(2.6)

We take the Lie derivative of the metric to confirm that these vector fields are in fact Killing
vector fields.
M > LieDerivative(KV, g);
0 du du, 0 du du, 0 du du, 0 du du, 0 du du

(2.7)

The electromagnetic field is not invariant under the isometry group of the spacetime. For
example,
M > LieDerivative(KV[2], F);
K2 e_r x r sin _r y C f u _r du Y dx K 2 e_r x r cos _r y C f u

_r du Y dy

(2.8)

The spacetime is a non-reductive homogeneous space.
The command SubspaceType computes the dimension and metric-induced signature of
the subspace of the tangent space which is spanned by the Killing vectors at a given point.
We see that the Killing vectors span the tangent space, so that the group they generate is
transitive and the spacetime is a homogeneous space.
M > SubspaceType(g, KV, [u = u0, v = v0, x = x0, y = y0])
assuming y0::real, y0 <> 0;
4, "PseudoRiemannian"

(2.9)

Next we compute the Lie algebra of the isometry group from the brackets of the Killing
vector fields and initialize the algebra (calling it "alg").
M > KValg := LieAlgebraData(KV, alg);
KValg := e1, e2 = e5, e1, e3 = _r e1, e1, e4 = Ke2, e3, e4 = _r e4, e3, e5 =
K_r e5
M > DGsetup(KValg):

(2.10)

The command IsotropySubalgebra finds a basis for the generators of the subgroup of the
isometry group which fixes a given point. Since the space is homogeneous, we can just
pick the origin of the chart without loss of generality. The output option exhibits the isotropy
subalgebra basis in terms of the chosen basis for alg.
M > h:=IsotropySubalgebra(KV, [u = 0, v = 0, x = 0, y = 0],
output=[alg]);
h d e1

(2.11)

A homogeneous space G / H with transitive group G and corresponding Lie algebra g is
called reductive if the isotropy subalgebra h admits an invariant complement m in g , that
is, g = h C m, and h, m 4 m . In this case h, m is called a reductive pair. We now
determine if this electrovac solution is a reductive homogeenous space. First construct the
general form of the complement of h with parameters (if any) labeled by t.
L1 > m := ComplementaryBasis(h, [e1, e2, e3, e4, e5], t);
m d t1 e1 C e2, t2 e1 C e3, t3 e1 C e4, t4 e1 C e5 , t1, t2, t3, t4

(2.12)

Now use the Query command to determine if this 4-parameter family of complements
includes an h - invariant complement.
alg > LieAlgebras:-Query(h, m, "ReductivePair")[1];
false

(2.13)

This spacetime is therefore a non-reductive homogeneous space.

Commands Illustrated
ComplementaryBasis, CovariantlyConstantTensors, EinsteinTensor,
EnergyMomentumTensor, HodgeStar, IsotropySubalgebra, KillingVectors, LieAlgebraData,
LieDerivative, MatterFieldEquations, PetrovType, Query, SubspaceType,
TensorInnerProduct,

Release Notes
• The illustrated commands are available in Maple 13 and subsequent releases.

